
1. OBJETIVOS

1.1  Objetivo General:

· Realizar un programa para las calculadoras de Texas Instruments (TI 68k) cuya finalidad es ayudar al estudiante que cursa la materia Economía para Ingenieros o Ingeniería Económica, en cuanto a agilizar los procesos de cálculo, documentándolo de manera adecuada y didáctica.

1.2  Objetivos Específicos:

· Hacer la programación en el lenguaje TI-Basic en Daisuke-Edit.

· Documentar el programa en cuanto a su forma de uso y de operación teniendo en  cuenta que el estudiante debe manejar los conceptos previo a su uso.

· Desarrollar unos ejercicios modelo para mostrar de forma práctica el uso de la herramienta.




2. INTRODUCCIÓN

 El desarrollo del aprendizaje de la materia economía para ingenieros, requiere de conceptos claves y dominio del tema. Pero igualmente se necesitan herramientas que disminuyan la pérdida de tiempo en los cálculos, a demás ilustrar y visualizar con mayor facilidad los tópicos concernientes a la materia.

Por eso se ha desarrollado un programa, el cual me facilite la determinación de cálculos, y visualización de procesos. Tenga en cuenta que el análisis de cada problema particular debe ser interpretado por el usuario.



















3. GENERALIDADES DEL PROCESO DE PROGRAMACIÓN
Se escribió el programa en Daisuke-Edit ejecutándolo en TIEmu. Se muestra un pantallazo del entorno de programación y el emulador.

[image: ]

El desarrollo del programa estuvo comprendido entre el 21 de febrero y el 3 de Marzo de 2008 y conlleva un proceso de prueba constante aprovechando la conectividad entre los dos programas antes mencionados.

Se programa de forma que no se generen errores de ningún tipo, ni en la eventualidad de que el usuario ingrese datos inválidos.

El programa es compatible con las calculadoras TI 68k de la Texas Instruments, las cuales son: TI-89, TI 92 (Puls), VoyageTM 200 y TI-89 Titanium. A continuación se muestran unas imágenes de las mismas en TIEmu.



[image: ]                                [image: ]
                   TI-89                                                                   TI-89 Titanium

[image: ]

TI-92 (Plus)

[image: ]
VoyageTM 200

4. CÓDIGO FUENTE DEL PROGRAMA

economía()
Prgm
© Fecha:   03-03-2008
© Título:  Ingeniería Económica
© Versión: 1.2
© Autor:   Froilán Andrés Morales Sanabria <froilan1986@hotmail.com>
©          http://www.ticalc.org/archives/files/authors/95/9551.html
©          UIS, Ingeniería Mecánica.
© Creado usando Daisuke-Edit http://www.daisuke.tk
Local fór,pf,ti,lvar,tx,ltx,nvar,dlogs,k,vark,var,vars,cont,desp,ex,ps,fs,is,ns,iis,ms,ias,as,gs,cs,ds,bs,vi,ve,mg,me,md,gr,ax,lb,pic
Try
 NewFold economía
Else
 setFold(economía)
EndTry
setMode("Graph","FUNCTION")→mg
setMode("Exact/Approx","APPROXIMATE")→me
setMode("Display Digits","FLOAT 12")→md
PlotsOff 
FnOff 
ClrDraw
setGraph("Grid","Off")→gr
setGraph("Axes","Off")→ax
setGraph("Labels","Off")→lb
{"p","f","i","n","ii","m","ia","a","g","c","d","b"}→lvar
For k,1,12
 lvar[k]→vark
 If getType(#vark)="NUM"
  string(#vark)→#(vark&"s")
EndFor
ClrIO
Loop
Dialog
 Title "INGENIERÍA ECONÓMICA v1.2"
 Text "Seleccione la Ecuación:"
 DropDown "  ",{"Interés Compuesto","Tasas","Anticipada-Vencida","Serie Uniforme","Grad. Aritmético","Grad. Geométrico","Grad. Escalonado","Flujos Combinados"},fór
 Text ""
 DropDown "Ver Imagen  ",{"No","Si"},vi
 DropDown "Ver Ecuación",{"No","Si"},ve
 Text ""
 Text " Por: Froilán A. Morales S."
 Text "  <froilan1986@hotmail.com>"
 Text "  UIS, Ingeniería Mecánica."
EndDlog
 If ok=0
  Exit
 If vi=2 Then
  If fór=1 Then
   "inte"→pic
  ElseIf fór=2 Then
   "tasa"→pic
  ElseIf fór=3 Then
   "vean"→pic
  ElseIf fór=4 Then
   "seru"→pic
  ElseIf fór=5 Then
   "arit"→pic
  ElseIf fór=6 Then
   "geom"→pic
  Else
   "esca"→pic
  EndIf
  If getType(#pic)="PIC" Then
   ClrDraw
   RclPic #pic
   Pause 
  EndIf
 EndIf
 If fór≥4 and fór≠8 Then
  0→pf
  PopUp {"Presente","Futuro"},pf
  If pf=0
   Cycle
 EndIf
 If fór=1 Then
  "INTERÉS COMPUESTO"→ti
  {"P","F","i","n"}→lvar
  "F=P*(1+i)^n"→tx
 ElseIf fór=2 Then
  "TASAS"→ti
  {"ii","i","m"}→lvar
  "i=(ii+1)^(1/m)-1"→tx
 ElseIf fór=3 Then
  "TASA ANTICIPADA"→ti
  {"i","ia"}→lvar
  "i=ia/(1-ia)"→tx
 ElseIf fór=4 Then
  "SERIE UNIFORME"→ti
  If pf=1 Then
   {"P","A","i","n"}→lvar
   "P=A*(1-(1+i)^¯n)/i"→tx
  Else
   {"F","A","i","n"}→lvar
   "F=A*((1+i)^n-1)/i"→tx
  EndIf
 ElseIf fór=5 Then
  "GRADIENTE ARITMÉTICO"→ti
  If pf=1 Then
   {"P","G","i","n"}→lvar
   "P=G*((1+i)^n-1-n*i)/(i^2*(1+i)^n)"→tx
  Else
   {"F","G","i","n"}→lvar
   "F=G*((1+i)^n-1-n*i)/i^2"→tx
  EndIf
 ElseIf fór=6 Then
  "GRADIENTE GEOMÉTRICO"→ti
  If pf=1 Then
   {"P","C","d","i","n"}→lvar
   {"P=C*(1-((1+d)/(1+i))^n)/(i-d)","P=n*C/(1+i)"}→ltx
  Else
   {"F","C","d","i","n"}→lvar
   {"F=C*((1+i)^n-(1+d)^n)/(i-d)","F=n*C*(1+i)^(n-1)"}→ltx
  EndIf
 ElseIf fór=7 Then
  "GRADIENTE ESCALONADO"→ti
  If pf=1 Then
   {"P","B","d","i","m","n"}→lvar
   {"P=B*((1+i)^m-1)/i*(1-((1+d)/(1+i)^m)^n)/((1+i)^m-1-d)","P=n*B*(1-(1+i)^¯m)/i"}→ltx
  Else
   {"F","B","d","i","m","n"}→lvar
   {"F=B*((1+i)^m-1)/i*((1+i)^(m*n)-(1+d)^n)/((1+i)^m-1-d)","F=n*B*(1-(1+i)^¯m)/i*(1+i)^(m*n)"}→ltx
  EndIf
 ElseIf fór=8 Then
  "FLUJOS COMBINADOS"→ti
   {"P","F","B","d","i","m","n"}→lvar
   {"P*(1+i)^(m*n)+B*((1+i)^m-1)/i*((1+i)^(m*n)-(1+d)^n)/((1+i)^m-1-d)+F=0","P*(1+i)^(m*n)+n*B*(1-(1+i)^¯m)/i*(1+i)^(m*n)+F=0"}→ltx
 EndIf
 If ve=2 Then
  If fór≤5 Then
   Pause tx
  Else
   If fór=8
    Disp "*Colocar signos adecuados"
   Pause ltx
  EndIf
 EndIf
 dim(lvar)→nvar
 ""→dlogs
 For k,1,nvar
  lvar[k]→vark
  dlogs&":Request """&vark&""","&vark&"s,0"→dlogs
 EndFor
 "Dialog:Title """&ti&""""&dlogs&":EndDlog"→dlogs
 Loop
  expr(dlogs)
  If ok=0
   Exit
  0→cont
  For k,1,nvar
   lvar[k]→var
   lvar[k]&"s"→vars
   If #vars="" Then
    DelVar #var
    cont+1→cont
    k→desp
   Else
    Try
     expr(#vars)→#var
    Else
     Pause "Error: Revise los valores ingresados ("&var&")."
     0→ok
     Exit
    EndTry
   EndIf
  EndFor
  If ok=0
   Cycle
  If cont=1 Then
   If fór=6 Then
    ltx[1]→tx
    If desp≠3 and desp≠4 Then
     If i=d Then
      ltx[2]→tx
     EndIf
    EndIf
   ElseIf fór=7 Then
    ltx[1]→tx
    If desp≠3 and desp≠4 and desp≠5 Then
     If (1+i)^m-1=d Then
      ltx[2]→tx
     EndIf
    EndIf
   ElseIf fór=8 Then
    ltx[1]→tx
    If desp≠4 and desp≠5 and desp≠6 Then
     If (1+i)^m-1=d Then
      ltx[2]→tx
     EndIf
    EndIf
   EndIf
   expr(tx)→ex
   lvar[desp]→var
   lvar[desp]&"s"→vars
   If fór≥4 and fór≠8 and desp≥3 Then
    If fór≥6 and desp=3 Then
     nSolve(ex,d)|d>¯1 and d<1→d
    Else
     nSolve(ex,#var)|#var>1.¯10→#var
    EndIf
   ElseIf fór=8 and desp≥4 Then
    If desp=4 Then
     nSolve(ex,d)|d>¯1 and d<1→d
    Else
     nSolve(ex,#var)|#var>1.¯10→#var
    EndIf
   Else
    right(solve(ex,#var))→#var
   EndIf
   string(#var)→#vars
  ElseIf cont=0 Then
   Pause "Error: Se debe dejar una variable indefinida (en blanco)."
  Else
   Pause "Error: Se admite únicamente una variable indefinida (en blanco)."
  EndIf
 EndLoop
EndLoop
setMode("Graph",mg)
PlotsOn 
FnOn 
setGraph("Grid",gr)
setGraph("Axes",ax)
setGraph("Labels",lb)
setMode("Exact/Approx",me)
setMode("Display Digits",md)
DispHome
EndPrgm











5. SOLUCIÓN DEL PRIMER EXAMEN PARCIAL
A modo de ejemplo se resolverán los siguientes ejercicios, para conocer la funcionalidad del programa.
Tenga en cuenta que el usuario debe realizar el análisis del problema y dar solución esquemática. Pues el programa solo hace operaciones, no planteamientos.

5.1  Ejercicio 1:
Universidad Industrial de Santander
Escuela de Estudios Industriales y Empresariales
Economía para Ingenieros - Primer examen parcial
Enero 29, 2008                                  Profesor: Carlos Enrique Vecino Arenas, Ph.D.

1. (Valor 40%)     Una empresa fabricante de calzado requiere remplazar varias máquinas para tecnificar su proceso productivo. Para su adquisición se tramita un préstamo de 100 millones de pesos a 10 años con un interés del 1.5% mensual, el cual deberá pagarse mediante una cuota mensual vencida que crece 2% cada mes respecto a la cuota anterior.
Según los cálculos del crédito, al final del quinto año, momento en el que la cuota mensual supera lo que la empresa tiene especificado como presupuesto mensual para el pago de créditos, el gerente financiero de la empresa decide hacer un abono extraordinario de $61’280.140,3 con el propósito de que la modalidad de pago pase a cuota fija.
 






5.1.1  Solución:

Análisis y Planteamiento del Problema:

La empresa fabricante de calzado en el instante del préstamo pensaba pagar sus cuotas siguiendo las condiciones iníciales del mismo. Por ello el esquema sería.


[image: ]


Este comportamiento es de un gradiente geométrico así que utilizando el programa de la calculadora Texas Instruments se deben seguir los siguientes pasos:




1. La calculadora debe trabajar en HOME.
[image: F:\Economía\Economía\Nuevos\screens\Inicio\screenshot000.bmp]

Estando allí se busca el programa en VAR-LINK, carpeta de economía, programa de economía, (La ubicación de la carpeta y nombre del programa son modificables) [ENTER]. Aparece algo así:
[image: F:\Economía\Economía\Nuevos\screens\Inicio\screenshot002.bmp]
Se cierra el paréntesis y de nuevo se da [ENTER]:
[image: F:\Economía\Economía\Nuevos\screens\Inicio\screenshot004.bmp]




A partir de éste momento el programa lo guiará realizando preguntas que permiten seleccionar el tipo correcto de ecuación, además si desea ver la gráfica para el tipo de problema, las variables y su significado en la ecuación, entre otras. 
Para la primera parte del problema se necesita un gradiente geométrico, para ello seleccionemos la ecuación adecuada. Si deseo ver la gráfica y la ecuación selecciono la opción si para cada una, (estas dos últimas opciones no son requisito para el desarrollo del programa) pero en cambio le permiten al usuario visualizar el tipo de problema a solucionar y los valores a reemplazar. Si le he asignado la opción de gráfica, la calculadora mostrará la siguiente imagen:

[image: F:\Economía\Economía\Nuevos\Archivos\Imágenes\Enviar\BMP\geom.bmp]

Oprimo [ENTER]:
[image: C:\Documents and Settings\CENTIC\Mis documentos\Mis imágenes\screens\Punto 1\screenshot000.bmp]
[image: F:\Economía\Economía\Nuevos\screens\Punto 1\screenshot001.bmp]
De nuevo [ENTER] y me da dos opciones Presente y Futuro. Esto se debe tener en cuenta dependiendo de las variables conocidas y la incógnita. En éste caso las variables conocidas son: d, i, P, n; y se desconocen F y C. Como necesito calcular C a partir de P → utilizo la fórmula de presente. 

[image: F:\Economía\Economía\Nuevos\screens\Punto 1\screenshot002.bmp]

Para un valor de: 


Para asignar valores en porcentajes se selecciona [CHAR] → Puntuation → %.
[image: F:\Economía\Economía\Nuevos\screens\Punto 1\screenshot003.bmp]
[image: F:\Economía\Economía\Nuevos\screens\Punto 1\screenshot004.bmp]
Una vez ingresados los valores, se deja en blanco el valor desconocido; (El programa solo admite un valor desconocido) y se presiona de nuevo la tecla [ENTER], en cuanto se ha despejado o calculado la incógnita se muestra de nuevo el cuadro de diálogo anterior pero con el valor de la incógnita.
[image: F:\Economía\Economía\Nuevos\screens\Punto 1\screenshot005.bmp]
  (Este sería el valor de la primera cuota y a partir de la cual se realizaran incrementos del 2%).
Como el préstamo se interrumpe al finalizar el quinto año, es decir en el final del periodo # 60. Se realiza un abono extraordinario y además se cambia la modalidad de pago a una cuota fija hasta el final del préstamo.

Gráfica real del préstamo

[image: ]
Transcurridos cinco años el saldo será: 
El valor futuro de la deuda sin abonos (–) el valor futuro de los abonos en ese mismo período 
→ Para el valor futuro de la deuda sin abonos se trabaja con la fórmula de interés compuesto en el programa. Con los valores de:
 
[image: F:\Economía\Economía\Nuevos\screens\Punto 1\screenshot010.bmp]
[image: F:\Economía\Economía\Nuevos\screens\Punto 1\screenshot011.bmp]
Siendo así   
El valor futuro de los abonos se calcula con la fórmula de gradiente geométrico.
→ Para valores de:
 
[image: F:\Economía\Economía\Nuevos\screens\Punto 1\screenshot006.bmp]  
[image: F:\Economía\Economía\Nuevos\screens\Punto 1\screenshot007.bmp]
El programa conserva los valores tanto ingresados como los calculados, en éste ejemplo ya habíamos calculado el valor de , por ello nos aparece en cuadro de diálogo su valor y no requiere ingresarlo de nuevo.
[image: F:\Economía\Economía\Nuevos\screens\Punto 1\screenshot008.bmp]
Oprimo [ENTER] y se calcula así el valor de F.
[image: F:\Economía\Economía\Nuevos\screens\Punto 1\screenshot009.bmp]



Realizando la diferencia entre F Y Fg obtengo:


La empresa decide abonar $61’280.140,3 para que la cuota no supere lo estipulado por la empresa mensualmente para pago de créditos, con este abono la deuda sería de:


El saldo será 

Para la nueva forma de pago ó nuevo sistema de crédito, (cuota fija) el saldo en ese instante sería el mismo presente.
Aplicando el programa para serie uniforme con valores de:




[image: F:\Economía\Economía\Nuevos\screens\Punto 1\screenshot012.bmp]

[image: F:\Economía\Economía\Nuevos\screens\Punto 1\screenshot013.bmp]
[image: F:\Economía\Economía\Nuevos\screens\Punto 1\screenshot014.bmp]
[image: F:\Economía\Economía\Nuevos\screens\Punto 1\screenshot015.bmp]
Cuando se suministran más datos de los requeridos el programa indica el error.
Obtengo el valor de la cuota fija que a su vez es la pregunta del problema.

















5.2  Ejercicio 2:
2. (Valor 60%)     Dado que la relación del precio de la pechuga de pollo respecto al de la pierna-pernil es superior en el exterior, una empresa avícola en Bucaramanga firma un contrato para exportar mensualmente 80 mil libras de pechuga durante 5 años. De acuerdo al estudio realizado se dispone de la siguiente información:
· La inversión requerida para la adecuación, implementación y puesta en marcha del proceso productivo y la logística de exportación asciende a $200’000.000 y toma poco tiempo (menos de un mes).
· Para la exportación se incurrirá en unos egresos mensuales incrementales, es decir, por encima de los costos que se tendrían si se vendiera las pechugas en Colombia. Dichos egresos incrementales por concepto de control de producción, documentación, trámite, embalaje, transporte y logística internacional están conformados por un monto fijo de $10’000.000 más un egreso de $500 por cada libra de pechuga.
· Se estima que todos los costos incrementales de producción y exportación crecerán 5% anualmente.
· Dado que la pechuga podría ser vendida en Colombia, obteniéndose un ingreso por libra de pechuga de 2 mil pesos que crecería anualmente un 5%, los ingresos por exportación deberían ser suficientes para cubrir tanto los costos mencionados previamente como estos ingresos que se dejarían de percibir.
· El cliente en el exterior depositará al momento de la exportación el valor pactado de “X” dólares por libra de pechuga y el banco de la empresa avícola convertirá dichos dólares a una tasa de cambio estimada de 2.000 pesos por dólar al momento de la primera exportación. Se estima que posteriormente la tasa de cambio crecerá mensualmente a un ritmo equivalente al 4,907021% anual.
Si se espera obtener mínimo 3% de rentabilidad mensual en la inversión en este proyecto, determine el valor mínimo por libra de pechuga “X” requerido para cubrir los ingresos dejados de recibir en Colombia, los costos incrementales (control de producción, embalaje y exportación) y para adicionalmente recuperar totalmente la inversión en 5 años (incluyendo obviamente la rentabilidad esperada).




5.2.1  Solución:

La idea para solucionar el problema es igualar todos los egresos generados por la exportación con los ingresos producidos por el pago de las pechugas en dólares, en el mismo instante de tiempo. Por eso se dividirá el problema analizando lo concerniente a la logística de exportación y luego los ingresos recibidos en la exportación.
Para la primera parte: (inversión inicial, costos de embalaje, logística, ingresos que se dejan de recibir en Colombia por libra de pechuga).

Gráfica de egresos para la exportación
[image: ]

Llevando todo al futuro del periodo 60.
Para el gradiente escalonado en rojo usando el programa con valores de:


[image: F:\Economía\Economía\Nuevos\screens\Punto 1\screenshot015.bmp]

[image: F:\Economía\Economía\Nuevos\screens\Punto 2\screenshot017.bmp]

[image: F:\Economía\Economía\Nuevos\screens\Punto 2\screenshot018.bmp]
[image: F:\Economía\Economía\Nuevos\screens\Punto 2\screenshot019.bmp]

Para el gradiente escalonado en negro con valores de:


[image: F:\Economía\Economía\Nuevos\screens\Punto 2\screenshot020.bmp]
[image: F:\Economía\Economía\Nuevos\screens\Punto 2\screenshot021.bmp]


Para la inversión inicial de 200’ el futuro por la ecuación de interés compuesto para valores de:



[image: F:\Economía\Economía\Nuevos\screens\Punto 2\screenshot022.bmp]
[image: F:\Economía\Economía\Nuevos\screens\Punto 2\screenshot023.bmp]
[image: F:\Economía\Economía\Nuevos\screens\Punto 2\screenshot024.bmp]


El futuro total que se espera obtener con la exportación de las pechugas es:





Para calcular el valor al cual se deben vender como mínimo las pechugas en el exterior en pesos. Se realizará mediante un gradiente geométrico. El incremento mensual equivalente al 4.907021% anual (de la tasa de cambio), será igual a: 
En este caso tenemos , y por ello utilizamos esta ecuación, para hallar , pero en la calculadora se toman como  e  ya que se usa la fórmula .
[image: F:\Economía\Economía\Nuevos\screens\Punto 2\screenshot025.bmp]

[image: F:\Economía\Economía\Nuevos\screens\Punto 2\screenshot026.bmp]

[image: F:\Economía\Economía\Nuevos\screens\Punto 2\screenshot027.bmp]









Para este gradiente geométrico con los valores de la gráfica mostrada a continuación, y usando el programa de la T.I. en la ecuación de gradiente geométrico:

[image: ]

[image: F:\Economía\Economía\Nuevos\screens\Punto 2\screenshot028.bmp]
[image: F:\Economía\Economía\Nuevos\screens\Punto 2\screenshot029.bmp]
[image: F:\Economía\Economía\Nuevos\screens\Punto 2\screenshot030.bmp]
[image: F:\Economía\Economía\Nuevos\screens\Punto 2\screenshot031.bmp]
Tengo como resultado un valor de 

Este valor es el equivalente al de 80.000 lbs. De pechuga por tanto el valor por libra de pechuga será:



La tasa de cambio en el mercado es de $ 2.000,00 / US$.

→ 






6. OBSERVACIONES Y CONCLUSIONES

· Cuando se desarrollan o se intentan resolver problemas se deben buscar medios que faciliten la solución, éste programa agiliza e ilustra problemas claves y básicos, que complementado con algunos conceptos de economía permite el desarrollo de cualquier problema planteado en ésta materia.

· El programa está diseñado para que el usuario interactué y navegue fácilmente. Es muy práctico y además indica los pasos a seguir.

· Se deja a consideración del profesor, la opción de adquirir el programa a través de un enlace en su portal web, ya que próximamente se subirá a la página www.tical.org en la lista de programas realizados por el integrante del grupo Froilán Morales en el siguiente enlace:
http://www.ticalc.org/archives/files/authors/95/9551.html

